HEIG-VD – 서울대 공대 SU Computer Science 2016
The Art of Compiler Construction

Course Project

The su (Summer University) Programming Language and Runtime Environment

Components

- **suPL**
  - a simple, C-like language
  - definition given as EBNF
- **suplc**
  - suPL compiler
  - written using Flex/Bison
  - **this is your job**
- **suVM**
  - a stack-based “computer” that can execute compiled suPL code
  - provided

Extended Backus-Naur Form

Extended Backus-Naur Form

- A formal notation to write down programming languages (context-free grammars)
  - meta symbols: “=” (definition), “,” (concatenation), “;” (end of production)

```
module   =   "module", ident, ";",
             { [typeDecl], [varDecl], [funcDecl] },
             "begin", stmtSeq, "end", ident, ".";
```

  - we use no symbol for concatenation, and “.” to indicate the end of the production

```
module   =   "module" ident ";"
             { [typeDecl] [varDecl] [funcDecl] }
             "begin" stmtSeq "end" ident ".".
```

Extended Backus-Naur Form

- Notation: Extended Backus-Naur Form

Notation	Usage	Example
=	definition	`letter = "A".."Z".`
.	termination	`letter = "A".."Z".`
|	alternation	`letter = "A".."Z" | "a".."z".`
[…]	option	`number = ["-"] digit.`
{ … }	repetition (≥ 0)	`number = ["-"] digit {digit}.`
(…)	grouping	`factor = [unaryOp] (ident | number).`
"…",'…'	terminal symbol	`"module", '"'`

The su Programming Language

The su Programming Language

- Simple C-based programming language
  - one data type: 32-bit signed integer
  - function calls (reentrant)
  - expression support
    - binary integer operations: +, -, *, /, %, ^
    - negation (“-a”) not supported
    - parentheses
  - control flow constructs
    - if – else
    - while
    - call – return
  - basic I/O
    - read/write
    - print string

suPL Example

```
int a, b;

int add(int p1, p2) {
  return p1 + p2;
}

void count(int N) {
  print "Counting to ";
  write N;

  int i;

  i = 0;
  while (i < N) {
    write i;
    i = i + 1;
  }
}

{
  int n;

  print "Enter n: ";
  read n;
  write add(1, n);
  count(n);
}
```

```
$ suplc test.su
$ suvm test.sux
Enter n: 5
6
Counting to 5
0
1
2
3
4
$
```

EBNF of suPL

```
program     =  decll stmtblock.

decll       =  { vardecl ';' | fundecl }.

vardecl     =  type ident { ',' ident }.

type        =  "integer" | "void".

fundecl     =  type ident '(' [ vardecl ] ')' stmtblock.

stmtblock   =  '{' { stmt } '}'.

stmt        =  vardecl ';' | assign | if | while | call ';' | return |
               read | write | print.
assign      =  ident '=' expression.

if          =  "if" '(' condition ')' stmtblock [ "else" stmtblock ].

while       =  "while" '(' condition ')' stmtblock.

call        =  ident '(' [ expression { ',' expression } ] ')'.

return      =  "return" [ expression ] ';'.
```

EBNF of suPL

```
read                =   "read" ident ';'.

write               =   "write" expression ';'.

print               =   "print" string ';'.


expression          =   number | ident |
                        expression '+' expression | expression '-' expression |
                        expression '*' expression | expression '/' expression |
                        expression '%' expression | expression '^' expression |
                        '(' expression ')' | call.
condition           =   expression "==" expression |
                        expression "<=" expression |
                        expression '<' expression.


number              =   DIGIT { DIGIT }.

ident               =   ALPHA { (ALPHA | DIGIT) }.

string              =   '"' { printable ASCII | "\t" | "\n" | '\"' | "\\" } '"'.
```

The suPL Compiler

- Flex/Bison-based compiler
  - you write the Flex/Bison files
  - helper functions and data structures provided
- Code generation on the fly
- Code format: binary

The su Virtual Machine

- Simulates a stack-based processor
  - a = b + 7

load b	load ‘b’ and push onto operand stack
push 7	push 7 onto operand stack
add	pop twice ($1, $2), compute $2+$1, push result
store a	pop value on top of stack, store into ‘a’

- Separate stack for variables & globals

Project Phase 1

suPL Scanner

suPL Scanner

- Prerequisites
  - Linux/POSIX system (MacOS should work)
    or
    VirtualBox development VM
- Your task: write a scanner (supl.lex) that correctly tokenizes suPL
- Hints and resources:
  - Flex manual: http://flex.sourceforge.net/manual/
  - identify each keyword separately (this will be necessary for the second phase)